ÜBUNG [00]:

Lies dir sämtliche Informationen über das Begleitpraktikum II auf der Homepage durch, damit du informiert über den Ablauf bist.
Beachte hierbei insbesondere die Fristen für die Anmeldung!
Hinweis: Falls du nicht im letzten Semester dein Mathestudium begonnen hast, so könnte der Abschnitt über Teilnehmende in der alten PO für dich relevant sein.

Euklidische Vektorräume

Aufgaben: 6

```
> restart;
  with(LinearAlgebra):
  with(plots):
```

Definitionen

MATH: Sei V ein $\mathbb{R}$-Vektorraum. Eine Abbildung $\Psi : V \times V \to \mathbb{R}$ heißt (positiv definites) **Skalarprodukt**, falls gilt:
1.) (*bilinear*) $\Psi(a \cdot v_1 + b \cdot v_2, w) = a \cdot \Psi(v_1, w) + b \cdot \Psi(v_2, w)$ für alle $a, b \in \mathbb{R}$ und $v_1, v_2, w \in V$ und
$\Psi(v, a \cdot w_1 + b \cdot w_2) = a \cdot \Psi(v, w_1) + b \cdot \Psi(v, w_2)$ für alle $a, b \in \mathbb{R}$ und $v, w_1, w_2 \in V$.
2.) (*symmetrisch*) $\Psi(v, w) = \Psi(w, v)$ für alle $v, w \in V$.
3.) (*positiv definit*) $\Psi(v, v) > 0$ für alle $v \in V$ mit $v \neq 0$.

Ist Ψ ein positiv definites Skalarprodukt auf dem endlich dimensionalen Vektorraum V, so heißt (V, Ψ) ein **Euklidischer Vektorraum** oder **endlich dimensionaler reeller Hilbertraum**. (Ist V nicht von endlicher Dimension, so spricht man von einem **Prähilbertraum**, einen **Hilbertraum** erhält man im unendlich-dimensionalen Fall durch Vervollständigung, ein Thema, das in der Analysis behandelt wird.)

MATH: Ein Vektor $v \in V$ heißt **normiert**, falls $\Psi(v, v) = 1$ gilt. Ist $v \in V$ mit $v \neq 0$ beliebig, so ist $\frac{1}{\sqrt{\Psi(v, v)}} \cdot v$ normiert. Wir nennen diesen Vorgang **normieren**. Er kann bei jedem Vektor $\neq 0$ in einem Euklidischen Vektorraum vorgenommen werden.

Wir können aber noch mehr: Wir können vermöge ψ zwei normierte Vektoren v, w miteinander vergleichen. Man schreibt dann $\Psi(v, w) =: \cos(\alpha)$ und nennt $\alpha \in [0, \pi]$ den **Winkel** zwischen v und w. Dass dies möglich ist, folgt aus der **Cauchy-Schwarzschen Ungleichung**, die wir unten beweisen werden.

MATH: Wir nennen zwei Vektoren v, w in V **orthogonal** (zueinander) oder **senkrecht** (aufeinander) **bezüglich** Ψ, falls $\Psi(v, w) = 0$ gilt. Eine Basis $B \in V^n$ von V heißt **Orthogonalbasis**, falls je zwei verschiedene Elemente B_i und B_j aufeinander senkrecht stehen. Falls die Basisvektoren noch zusätzlich normiert sind, spricht man von einer **Orthonormalbasis** (kurz ONBasis).

MATH: Ist $B \in V^n$ eine Basis von V und $\Psi : V \times V \to \mathbb{R}$ eine Bilinearform, so heißt die Matrix $G \in \mathbb{R}^{n \times n}$ mit $G_{i,j} = \Psi(B_i, B_j)$ die **Grammatrix** bezüglich B.
Im Falle einer linearen Abbildung haben wir die Abbildungsmatrix kennengelernt. Ein ähnlicher Fall liegt hier vor:
Die Grammatrix ermöglicht es uns, das Skalarprodukt zweier Vektoren v, $w \in V$ durch ihre Koordinatenspalten und Matrixmultiplikationen auszurechnen:
$\Psi(v, w) = ({}^{B}v)^{tr} \cdot G \cdot {}^{B}w$, falls G die Grammatrix bezüglich B ist.

Jeder Euklidische Vektorraum lässt eine Basis zu, bezüglich der die Grammatrix die Einheitsmatrix ist. Eine Basis ist genau von dieser Form, wenn sie eine ONBasis ist.

BEISPIEL: $\mathbb{R}^4$ mit dem **Standardskalarprodukt** ist ein Euklidischer Vektorraum:

```
> Vector[row](4,symbol=a);
  Vector[row](4,symbol=b);
```

$$\begin{bmatrix} a_1 & a_2 & a_3 & a_4 \end{bmatrix}$$

$$\begin{bmatrix} b_1 & b_2 & b_3 & b_4 \end{bmatrix} \quad \textbf{(1.1.1)}$$

Das Standardskalarprodukt zweier Vektoren aus $\mathbb{R}^4$ ist definiert durch:

```
> BilinearForm(Vector[row](4,symbol=a),Vector[row](4,symbol=
  b),conjugate=false);
```

$$a_1\, b_1 + a_2\, b_2 + a_3\, b_3 + a_4\, b_4 \quad \textbf{(1.1.2)}$$

Wir betrachten die Standardbasis S

```
> S:=[<1 | 0 | 0 | 0>,<0 | 1 | 0 | 0>,<0 | 0 | 1 | 0>,<0 | 0 |
  0 | 1>];
```

$$S := \left[\begin{bmatrix} 1 & 0 & 0 & 0 \end{bmatrix}, \begin{bmatrix} 0 & 1 & 0 & 0 \end{bmatrix}, \begin{bmatrix} 0 & 0 & 1 & 0 \end{bmatrix}, \begin{bmatrix} 0 & 0 & 0 & 1 \end{bmatrix}\right] \quad \textbf{(1.1.3)}$$

und die zugehörige Grammatrix ${}_{S}\Psi^{S}$:

```
> Matrix(4,4,(i,j)-> BilinearForm(S[i],S[j],conjugate=false));
```

$$\begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix} \quad \textbf{(1.1.4)}$$

Wir betrachten eine andere Basis B:

```
> B:=[<1 | 1 | 1 | 1>,<1 | 1 | 1 | 2>,<1 | 1 | 0 | 3>,<1 | 0 |
```

```
0 | 4>];
```

$$B := [[1\ 1\ 1\ 1], [1\ 1\ 1\ 2], [1\ 1\ 0\ 3], [1\ 0\ 0\ 4]] \quad (1.1.5)$$

Die Grammatrix ${}_B\Psi^B$ bezüglich dieser Basis:

```
> G:=Matrix(4,4,(i,j)-> BilinearForm(B[i],B[j],conjugate=
  false));
```

$$G := \begin{bmatrix} 4 & 5 & 5 & 5 \\ 5 & 7 & 8 & 9 \\ 5 & 8 & 11 & 13 \\ 5 & 9 & 13 & 17 \end{bmatrix} \quad (1.1.6)$$

MATH: Sind B und C Basen, und ${}_B\Psi^B$, ${}_C\Psi^C$ die zugehörigen Grammatrizen. Dann gilt ${}_B\Psi^B = ({}^C Id^B)^{tr}\, {}_C\Psi^C\, {}^C Id^B$. Mit ${}_B Id_C := ({}^C Id^B)^{tr}$ schreiben wir dann ${}_B\Psi^B = {}_B Id_C\, {}_C\Psi^C\, {}^C Id^B$.

ÜBUNG [01]:

Es sei $V := \mathbb{R}^{2 \times 2}$ mit $\Psi : V \times V \to \mathbb{R} : (A, B) \mapsto Spur\left(A \cdot B^{tr}\right)$ ein euklidischer Vektorraum.

1.) Bestimme eine Basis B von V und damit ${}_B\Psi^B$.

2.) Bestimme den Winkel zwischen den beiden folgenden Matrizen:

```
> A1:=Matrix([[1,2],[3,4]]);
  A2:=Matrix([[7,1],[5,-2]]);
```

$$A1 := \begin{bmatrix} 1 & 2 \\ 3 & 4 \end{bmatrix}$$

$$A2 := \begin{bmatrix} 7 & 1 \\ 5 & -2 \end{bmatrix} \quad (1.1.7)$$

3.) Bestimme den Winkel zwischen den beiden folgenden Matrizen:

```
> B1:=Matrix([[1,2],[2,1]]);
  B2:=Matrix([[0,-5],[5,0]]);
```

$$B1 := \begin{bmatrix} 1 & 2 \\ 2 & 1 \end{bmatrix}$$

$$B2 := \begin{bmatrix} 0 & -5 \\ 5 & 0 \end{bmatrix} \quad (1.1.8)$$

Konstruktion einer Orthonormalbasis

MATH: Es gilt: Ist $B \in V^n$ eine Basis von V, so gibt es eine ONBasis O von V mit

$$\langle B_1, \ldots, B_k \rangle = \langle \mathrm{O}_1, \ldots, \mathrm{O}_k \rangle$$

für $k = 1, \ldots, n$.

Bei der Konstruktion dieser ONBasis hat man nur wenige Möglichkeiten: O_1 ist die Normierung von B_1 oder das Negative hiervon, wir entscheiden uns für

$$\mathrm{O}_1 = \frac{1}{\sqrt{\Psi(B_1, B_1)}} B_1$$

Um O_2 zu bekommen, setzen wir

$$\mathrm{O}'_2 = a \cdot \mathrm{O}_1 + B_2$$

und bestimmen a so, dass O'_2 senkrecht auf O_1 steht:

$$0 = \Psi(\mathrm{O}_1, \mathrm{O}'_2) = a \cdot \Psi(\mathrm{O}_1, \mathrm{O}_1) + \Psi(\mathrm{O}_1, B_2) = a + \Psi(\mathrm{O}_1, B_2).$$

Wir setzen also $a = -\Psi(\mathrm{O}_1, B_2)$ und bekommen Orthogonalität. Anschließend normieren wir O'_2 und erhalten O_2.

Um O_3 zu bekommen, setzen wir

$$\mathrm{O}'_3 = a \cdot \mathrm{O}_1 + b \cdot \mathrm{O}_2 + B_3$$

und bestimmen auf deselbe Weise a und b so, dass O'_3 senkrecht auf O_1 und O_2 steht. Anschließend normieren wir wieder, um O_3 zu bekommen. Dann fahren wir auf dieselbe Weise mit $\mathrm{O}_4, \ldots, \mathrm{O}_n$ und haben eine ONBasis bekommen.

Dieses Verfahren heißt auch **Gram-Schmidtsches Orthogonalisierungsverfahren** oder **Gram-Schmidtsches Orthonormalisierungsverfahren**.

DENKANSTOSS: Es sind genau 2^n Wahlen für die ONBasis mit der obigen Eigenschaft möglich.

ÜBUNG [02]:

Betrachte den $\mathbb{R}^4$ mit dem Standardskalarprodukt und folgender Basis

```
> B:=[<1 | 1 | 1 | 1>,<1 | 1 | 1 | 0>,<1 | 1 | 0 | 0>,<1 | 0 
  | 0 | 0>];
```

$$B := \left[\left[\begin{matrix} 1 & 1 & 1 & 1 \end{matrix} \right], \left[\begin{matrix} 1 & 1 & 1 & 0 \end{matrix} \right], \left[\begin{matrix} 1 & 1 & 0 & 0 \end{matrix} \right], \left[\begin{matrix} 1 & 0 & 0 & 0 \end{matrix} \right] \right] \quad \textbf{(1.2.1)}$$

1) Bestimme die Grammatrix **G** bezüglich **B**.

2) Bestimme nach dem obigen Verfahren eine ON-Basis **C** von der Basis **B**.

3) Wie muss man eine Basiswechselmatrix zwischen den Basen **B** und **C** anwenden, damit man die zu einer ON-Basis passende Einheitsmatrix erhält?

MATH: Man sieht, dass man aus dem Gram-Schmidtschen Verfahren auch dann eine Orthonormalbasis bekommt, wenn man nur mit einem endlichen Erzeugendensystem B (und nicht unbedingt mit einer Basis) anfängt.

Winkel, Orthogonalprojektionen, Approximationen

MATH: Als Folgerung bekommen wir die **Cauchy-Schwarzsche Ungleichung**:

$$\Psi(v, v)\cdot\Psi(w, w) - \Psi(v, w)^2 \geq 0$$

mit Gleichheit genau dann, wenn v und w linear abhängig sind.

Beweis:
Wir können von einer ONBasis (B_1, B_2) von $\langle v, w\rangle$ ausgehen, so dass

$$v = a\cdot B_1$$
$$w = b\cdot B_1 + c\cdot B_2.$$

Die linke Seite ist dann

$$a^2\cdot(b^2 + c^2) - (a\cdot b)^2 = a^2\cdot c^2$$

und die Behauptung folgt.

Insbesondere ist der Winkel zwischen zwei Vektoren v, w, die beide von Null verschieden sind, wohldefiniert.

Wir betrachten den Fall zweier Einheitsvektoren $v = (1, 0)$ und
$w = \left(\cos\left(\frac{\pi}{3}\right), \sin\left(\frac{\pi}{3}\right)\right)$

```
> d1:=plot([cos(t),sin(t),t=0..2*Pi]):
  d2:=plot([cos(Pi/3),t*sin(Pi/3),t=0..1]):
  d3:=plot([t*cos(Pi/3),t*sin(Pi/3),t=0..1]):
  display([d1,d2,d3], scaling=constrained);
```



Wir sehen: Die **Orthogonalprojektion** von w auf $\langle v \rangle$ (entlang dem zu $\langle v \rangle$ senkrechten Teilraum) ist $\Psi(v, w) \cdot v$.

```
> BilinearForm(<1,0>,<cos(Pi/3),sin(Pi/3)>,conjugate=false)*
  <1,0>;
```

$$\begin{bmatrix} \frac{1}{2} \\ 0 \end{bmatrix} \tag{1.3.1}$$

Zur genauen Definition der Orthogonalprojektion siehe unten.

MATH: Mit dieser geometrischen Interpretation des Skalarproduktes können wir das Gram-Schmidtsche Verfahren geometrisch sehr leicht verstehen: Durch Subtraktion der Orthogonalprojektion von B_2 auf O_1 erhalten wir einen Vektor, der senkrecht auf O_1 steht, etc.

BEISPIEL: Welche Polynomfunktion f vom Grad ≤ 3 der Form $x \mapsto x^3 + a \cdot x^2 + b \cdot x + c$ liefert einen minimalen Wert für

$$\sum_{i=0}^{4} f(i)^2?$$

Die Basis

```
> B:=[1,x,x^2];
```

$$B := [1, x, x^2] \tag{1.3.2}$$

des Teilraumes der Polynomfunktionen vom Grad < 3 wird zuerst in eine ON-Basis verwandelt: Um die Quadratsummen ausdrücken zu können nehmen wir als Skalarprodukt

```
> SkPr:=proc(f::polynom,g::polynom)
>   return add(subs(x=i,f*g),i=0..4);
> end proc:
```

und benutzen das Gram-Schmidtsche Orthogonalisierungsverfahren:

```
> a:='a':b:='b':c:='c':
> o[1]:=B[1];
```

$$o_1 := 1 \tag{1.3.3}$$

```
> o[1]:=simplify(1/sqrt(SkPr(o[1],o[1]))*o[1]);
```

$$o_1 := \frac{1}{5}\sqrt{5} \tag{1.3.4}$$

```
> o[2]:=a*o[1]+B[2];
```

$$o_2 := \frac{1}{5}\, a\sqrt{5} + x \tag{1.3.5}$$

```
> o[2]:=subs(isolate(SkPr(o[1],o[2]),a),o[2]);
```

$$o_2 := -2 + x \tag{1.3.6}$$

```
> o[2]:=simplify(1/sqrt(SkPr(o[2],o[2]))*o[2]);
```

$$o_2 := \frac{1}{10}\sqrt{10}\,(-2 + x) \tag{1.3.7}$$

```
> o[3]:=a*o[1]+b*o[2]+B[3];
```

$$o_3 := \frac{1}{5}\, a\sqrt{5} + \frac{1}{10}\, b\sqrt{10}\,(-2 + x) + x^2 \tag{1.3.8}$$

```
> o[3]:=subs(isolate(SkPr(o[1],o[3]),a),o[3]);
```

$$o_3 := -6 + \frac{1}{10}\, b\sqrt{10}\,(-2 + x) + x^2 \tag{1.3.9}$$

```
> o[3]:=subs(isolate(SkPr(o[2],o[3]),b),o[3]);
```

$$o_3 := x^2 - 4\,x + 2 \tag{1.3.10}$$

```
> o[3]:=simplify(1/sqrt(SkPr(o[3],o[3]))*o[3]);
```

$$o_3 := \frac{1}{14}\sqrt{14}\,(x^2 - 4\,x + 2) \tag{1.3.11}$$

Der Teilraum $\langle 1, x, x^2 \rangle$ enthält eine Lösung des Problems, nämlich 0. Wir suchen das Element der Form $n := x^3 + a \cdot x^2 + b \cdot x + c$, das im Sinne unseres Skalarprodukts "am nächsten" an diesem Teilraum liegt. Dazu suchen wir das eindeutige n, das senkrecht auf diesem Teilraum steht:

```
> n:=a*o[1]+b*o[2]+c*o[3]+x^3;
```

$$n := x^3 + \frac{1}{5}\, a \sqrt{5} + \frac{1}{10}\, b \sqrt{10}\, (-2 + x) + \frac{1}{14}\, c \sqrt{14}\, (x^2 - 4\,x + 2) \qquad \textbf{(1.3.12)}$$

```
> n:=subs(isolate(SkPr(o[1],n),a),n);
```

$$n := x^3 - 20 + \frac{1}{10}\, b \sqrt{10}\, (-2 + x) + \frac{1}{14}\, c \sqrt{14}\, (x^2 - 4\,x + 2) \qquad \textbf{(1.3.13)}$$

```
> n:=subs(isolate(SkPr(o[2],n),b),n);
```

$$n := x^3 + \frac{54}{5} - \frac{77}{5}\, x + \frac{1}{14}\, c \sqrt{14}\, (x^2 - 4\,x + 2) \qquad \textbf{(1.3.14)}$$

```
> n:=subs(isolate(SkPr(o[3],n),c),n);
```

$$n := x^3 - \frac{6}{5} + \frac{43}{5}\, x - 6\, x^2 \qquad \textbf{(1.3.15)}$$

```
> n := -6/5+43/5*x-6*x^2+x^3;
```

$$n := x^3 - \frac{6}{5} + \frac{43}{5}\, x - 6\, x^2 \qquad \textbf{(1.3.16)}$$

```
> map(i->SkPr(n,o[i]), [$1..3]);
  map(i->SkPr(n,x^i), [$0..2]);
```

$$[0, 0, 0]$$

$$[0, 0, 0] \qquad \textbf{(1.3.17)}$$

```
> SkPr(n,n);
```

$$\frac{72}{5} \qquad \textbf{(1.3.18)}$$

```
> plot([x^3,n],x=0..4,color=[red,green]);
```



Man könnte $\sqrt{\mathrm{SkPr}(n,\ n)}$ als den **Abstand** von x^3 von dem Teilraum $\langle 1,\ x,\ x^2\rangle$ bezeichnen. Ebenso könnte man $x^3 - n$, also das Bild der Orthogonalprojektion, als eine **Approximation** von x^3 durch Polynome vom Grad < 3 *im Sinne unseres Skalarproduktes* ansprechen:

```
> plot([x^3,x^3-n],x=0..4,color=[red,blue]);
```



MATH: Haben wir eine Orthonormalbasis $B \in V^n$ unseres Euklidischen Vektorraumes (V, Ψ), so gilt für alle $v \in V$:

$$v = \Psi(v, B_1) \cdot B_1 + \ldots + \Psi(v, B_n) \cdot B_n$$

Die $\Psi(v, B_i)$ heißen auch **Fourier-Koeffizienten** von v.

MATH: Was wir aus der obigen Formel kennenlernen, sind **Orthogonalprojektionen**: Ist W ein Teilraum von V und B in V^n eine Orthonormalbasis von V, so dass $(B_1, \ldots, B_k)$ eine Basis (und damit Orthonormalbasis) von W ist, so heißt

$$\pi : V \to V : v \mapsto \Psi(v, B_1) \cdot B_1 + \ldots + \Psi(v, B_k) \cdot B_k$$

die **Orthogonalprojektion** von V auf W.

Dass Orthogonalprojektionen in der Tat Projektionen sind, können wir leicht nachrechnen:

ÜBUNG [03]:
Zeige: Die Orthogonalprojektion ist eine Projektion (d.h. ist π eine Orthogonalprojektion, so gilt $\pi^2 = \pi$).

MATH: Orthogonalprojektionen sind charakterisiert als die Projektionen, bei denen Kern und Bild orthogonal zueinander sind.

ÜBUNG [04]:
Zeige: Die Definition der Orthogonalprojektion ist unabhängig von der gewählten Basis B. *Hinweis*: Zeige für eine weitere "passende" ONB C gilt: $({}^{C}Id^{B})^{-1} = ({}^{C}Id^{B})^{tr}$ und teile die beteiligten Matrizen in Blöcke auf.

MATH: Beachte: Eine Orthogonalprojektion ist bereits durch ihren Bildraum festgelegt.

Um ein Gefühl für Orthogonalprojektionen zu bekommen, schauen wir uns ein Beispiel an, mit dem wir bereits vertraut sind:

ÜBUNG [05]:
1) (Wiederholung) Gib alle Projektionen von $\mathbb{R}^{3\times 1}$ (durch ihre Matrizen bezüglich der Standardbasis) an, deren Bild von $(1, 1, 1)^{tr}$ erzeugt wird. 2) Welche von diesen ist die Orthogonalprojektion, wenn wir die Standardbasis als Orthonormalbasis nehmen? (Erinnerung: Der Kern der Projektion ist das Komplement des Bildes.)

DENKANSTOSS: Warum sind die Matrizen von Orthogonalprojektionen bezüglich ONBasen immer symmetrisch, d. h. gleich ihren Transponierten?

MATH: Jedenfalls sieht man auch leicht aus der Formel

$$v = \Psi(v, B_1)\cdot B_1 + \ldots + \Psi(v, B_n)\cdot B_n,$$

dass das Bild unter der Orthogonalprojektion $\pi(v)$ das eindeutige Element von $W = \langle B_1,\ldots, B_k\rangle$ ist, welches am nächsten an v liegt, für das also $\| v - \pi(v)\|$

$= \sqrt{\Psi(v-\pi(v), v-\pi(v))}$ minimal unter allen Elementen aus W ist. Dieser Wert heißt auch der **Abstand** von v von W und $\pi(v)$ die **beste Approximation** von v an W.

Haben wir also erstmal die Orthogonalprojektion von V auf W gefunden, so macht sie es uns sehr leicht, die beste Approximation eines Vektors v an W auszurechnen:

ÜBUNG [06]:
Bestimme (unter Benutzung der Lösung der letzten Aufgabe) die beste

Approximation und den Abstand von $\begin{bmatrix} 1 \\ 0 \\ 0 \end{bmatrix}$ zu dem Raum $\left\langle \begin{bmatrix} 1 \\ 1 \\ 1 \end{bmatrix} \right\rangle$, wobei die Standardbasis Orthonormalbasis ist.

Spektralsatz

Aufgaben: 3

```
> restart;
  with(LinearAlgebra):
  with(plots):
```

Der Spektralsatz

MATH: Ein Euklidischer Vektorraum (V, Ψ), wo V ein endlichdimensionaler reeller Vektorraum ist und

$$\Psi : V \times V \rightarrow \mathbb{R}$$

ein positiv definites Skalarprodukt, hat viele Endomorphismen, die in der einen oder anderen Weise auf das Skalarprodukt abgestimmt sind. An erster Stelle seien die Orthogonalprojektionen genannt. Eine Verallgemeinerung davon sind die selbstadjungierten Endomorphismen. Wir beginnen mit den einfachsten Fällen.

MATH: Eine Projektion von V ist gerade ein Endomorphismus π von V mit $\pi^2 = \pi$. In diesem Fall zerlegt sich V als direkte Summe von $Kern(\pi)$ und $Bild(\pi) = Kern(Id_V - \pi)$. Umgekehrt liefert jede Zerlegung von V in eine direkte Summe von zwei Teilräumen ein Paar von Projektionen π_1 und π_2, die die Teilräume als Bilder oder Kerne reproduzieren. Wichtig ist, dass man tatsächlich beide Teilräume braucht, um die Projektionen festzulegen.

MATH: Das Komplement (bezüglich der direkten Summe) eines Teilraumes ist i. A. nicht eindeutig gegeben. Man konnte aber das Skalarprodukt des Euklidischen Vektorraumes benutzen, um ein eindeutiges Komplement zu erhalten. Man verlangt bei einer **Orthogonalprojektion** zusätzlich zu der Projektionseigenschaft, dass die Vektoren aus Kern und Bild senkrecht aufeinander stehen, also Skalarprodukt Null haben. Vorstellungsmäßig erreicht man dies am einfachsten dadurch, dass man eine Orthonormalbasis B des Bildes, also des Teilraumes, auf den man projizieren will, zu einer Orthonormalbasis des ganzen Raumes C ergänzt, und die neuen Basisvektoren zu einer Orthonormalbasis D des Kernes wählt, also des Teilraumes, entlang dessen man projiziert.
Ist B eine Orthonormalbasis von V und $(B_1,.., B_k)$ eine Basis des Teilraumes W, so

wird die Orthogonalprojektion von V auf W bezüglich B durch die Diagonalmatrix $Diag([1,.., 1, 0,.., 0])$ beschrieben, wo wir k Einsen und $n-k$ Nullen haben.

MATH: In einem Euklidischen Vektorraum (V, Ψ) definiere für den Endomorphismus α die **Adjungierte** α^{ad} über:

$$\Psi\left(\alpha^{ad}(v), w\right) = \Psi(v, \alpha(w))$$

für alle $v, w \in V$.

MATH: Die Matrix T einer Basistransformation von einer Orthonormalbasis in eine andere ist orthogonal, d. h. $T^{tr}.T = I$, also ist die Transponierte gleich der Inversen.

MATH: Als Folgerung erhalten wir, dass die Matrix M einer Orthonormalprojektion π bezüglich irgendeiner Orthonormalbasis immer symmetrisch ist: $M = T^{-1}Diag([1,.., 1, 0,..0])T = T^{tr}Diag([1,.., 1, 0,..0])T = M^{tr}$.
Was wir jetzt in Matrizen nachgerechnet haben, bedeutet für die lineare Abbildung, dass π selbstadjungiert ist, also gleich seiner Adjungierten π^{ad}.
Umgekehrt sind selbstadjungierte Projektionen Orthogonalprojektionen.

MATH (Spektralsatz): Sei (V, Ψ) ein Euklidischer Vektorraum und α ein selbstadjungierter Endomorphismus von V. Dann existiert eine Orthonormalbasis aus Eigenvektoren von α.
Matrixversion: Ist $A \in \mathbb{R}^{n \times n}$ eine symmetrische Matrix, so existiert eine orthogonale Matrix $g \in \mathbb{R}^{n \times n}$ mit $g^{tr} \cdot A \cdot g = g^{-1} \cdot A \cdot g$ diagonal.

Selbstadjungierte Endomorphismen α unseres Euklidischen Vektorraumes (V, Ψ), besitzen also ein vielfachheitenfreies (**DENKANSTOSS:** Warum?) Minimalpolynom, welches über = in Linearfaktoren zerfällt (**DENKANSTOSS:** Warum?); sagen wir

$$\mu(x) = \prod_{i=1}^{k} (x - a_i) \text{ mit } a_i \in \mathbb{R}.$$

Setzt man in ein Polynom eine symmetrische Matrix ein, bekommt man wieder eine symmetrische Matrix. Insbesondere sind die Projektionen auf Eigenräume, die man als solche Matrixpolynome bekommt, selbstadjungiert, also Orthogonalprojektionen. Wählt man in den Eigenräumen ON-Basen, hat man eine ON-Basis aus Eigenvektoren.

ÜBUNG [07]:

1) Führe die gerade diskutierte Zerlegung am Beispiel der folgenden Matrix A durch, d.h. bestimme die Projektionen auf die Eigenräume.
(Hierbei wollen wir die Matrix A als Matrix eines (nicht näher spezifizierten) Endomorphismus bezüglich einer (nicht näher spezifizierten) ONBasis auffassen.)
2) Warum sind diese Projektionen durch symmetrische Matrizen gegeben?

3) Bestimme eine ON-Basis aus Eigenvektoren von A

```
> A:=Matrix(4,4,(i,j)->1)+1;
```

$$A := \begin{bmatrix} 2 & 1 & 1 & 1 \\ 1 & 2 & 1 & 1 \\ 1 & 1 & 2 & 1 \\ 1 & 1 & 1 & 2 \end{bmatrix} \tag{2.1.1}$$

Der Spektralsatz hat viele Anwendungen. Wir wollen ein Beispiel betrachten.

Spektralsatz und Kegelschnitte

Der Spektralsatz hat noch eine wichtige Konsequenz, die man benutzen kann, um die Diskussion gewisser geometrischer Gebilde zu führen.

MATH: Sei $p = \sum_{i,\,j=1}^{n} a_{ij} x_i x_j$ ein quadratisches und homogenes Polynom in den n Variablen $x_1,..,x_n$, so gibt es eine orthogonale Variablensubstitution $y_i = \sum_{j=1}^{n} T_{ij} \cdot x_j$, so dass das transformierte Polynom als $\sum_{i=1}^{n} r_i \cdot y_i^2$ geschrieben werden kann mit (bis auf Reihenfolge) eindeutig bestimmten reellen Zahlen $r_1,..., r_n$. Orthogonal heißt hierbei, dass die Matrix T orthogonal ist.
Zum Beweis mache man sich klar, dass man ohne Einschränkung der Allgemeinheit $a_{ij} = a_{ji}$ annehmen kann, also eine symmetrische Koeffizientenmatrix. Deren Spektralzerlegung liefert das gewünschte Ergebnis.

Beispiel: Symmetrische Matrizen aus $\mathbb{R}^{n \times n}$

```
> A := Matrix(2,2,[[17/50,3/25],[3/25,41/100]]);
```

$$A := \begin{bmatrix} \frac{17}{50} & \frac{3}{25} \\ \frac{3}{25} & \frac{41}{100} \end{bmatrix} \tag{2.2.1}$$

und reelle homogene Polynome von Grad 2 in n Unbekannten stehen in Bijektion zueinander:

DENKANSTOSS: Gib die Bijektion und ihre Inverse an.

```
> p:=expand(Transpose(Vector(i->x[i],1..2)).A.Vector(i->x[i],
  1..2));
```

$$p := \frac{17}{50} x_1^2 + \frac{6}{25} x_1 x_2 + \frac{41}{100} x_2^2 \tag{2.2.2}$$

Nach dem Spektralsatz kann man eine symmetrische Matrix durch einen orthogonalen Basiswechsel diagonalisieren:

```
> E:=Eigenvectors(A):
  if E[1][1]<E[1][2] then # wir wollen eine bestimmte
  Reihenfolge der Eigenwerte
    T:=E[2]:
  else
    T:=SubMatrix(E[2],1..2,[2,1]):
  fi:
  T:=evala(T.DiagonalMatrix(simplify(map(i->sqrt(evala
  (Transpose(T).T)[i,i]),[$1..2])))^(-1));
```

$$T := \begin{bmatrix} -\frac{4}{5} & \frac{3}{5} \\ \frac{3}{5} & \frac{4}{5} \end{bmatrix} \tag{2.2.3}$$

```
> evala(T^(-1).A.T);
```

$$\begin{bmatrix} \frac{1}{4} & 0 \\ 0 & \frac{1}{2} \end{bmatrix} \tag{2.2.4}$$

```
> evala(Transpose(T).T); # T ist wirklich orthogonal
```

$$\begin{bmatrix} 1 & 0 \\ 0 & 1 \end{bmatrix} \tag{2.2.5}$$

Man kann den selben Basiswechsel auf das Polynom anwenden. (**DENKANSTOSS**: Dies ist nichts anderes als ein Koordinatenwechsel.) Dies liefert ein Polynom ohne gemischte Terme:

```
> l:=evala(simplify(solve(convert(Vector(i->y[i],1..2)-T^(-1).
  Vector(i->x[i],1..2),list),[x[1],x[2]])[1])):
  q:=collect(simplify(expand(subs(l,p))),[y[1],y[2]],evala);
```

$$q := \frac{1}{2}\, y_2^2 + \frac{1}{4}\, y_1^2 \tag{2.2.6}$$

Warum ist eine Transformation von Polynomen auf solche eine Form interessant? Wir haben einen besseren Überblick. Wir schauen uns dies am Beispiel von Ellipsen an.

MATH: Eine **Ellipse** ist der geometrische Ort aller Punkte $P \in \mathbb{R}^2$, deren Abstandssumme von zwei festen Punkten B_1, $B_2 \in \mathbb{R}^2$ einen festen Wert k hat. Diese beiden Punkte nennt man **Brennpunkte**.

Behauptung: Betrachten wir nun von obigem q die Faser $q^{-1}(\{1\})$, so ist dies eine Ellipse mit den beiden Brennpunkten

```
> B1:=Vector([sqrt(2),0]);
  B2:=Vector([-sqrt(2),0]);
```

$$B1 := \begin{bmatrix} \sqrt{2} \\ 0 \end{bmatrix}$$

$$B2 := \begin{bmatrix} -\sqrt{2} \\ 0 \end{bmatrix} \tag{2.2.7}$$

und dem Wert

```
> k:=4;
```

$$k := 4 \tag{2.2.8}$$

Wir zeichnen die Faser der 1 und es scheint sich wirklich um eine Ellipse zu handeln.

```
> with(plots):
  implicitplot(q-1,y[1]=-3..23,y[2]=-3..3,color=blue,scaling=
  constrained,grid=[300,300]);
```




```
> ell:=implicitplot(q-1,y[1]=-3..3,y[2]=-3..3,color=blue,
  scaling=constrained,grid=[300,300]):
```

```
brp:=pointplot(map(convert,[B1,B2],list),color=red,symbol =
solidcircle,symbolsize=20):
display(ell,brp);
```



Wir betrachen als ersten Test zwei Punkte der Faser der 1.

```
> v1:=Vector([sqrt(2),1]);
'q(v1)'=subs(map(i->y[i]=v1[i],[1,2]),q);
'Abstandssumme'=sqrt(BilinearForm(v1-B1,v1-B1,conjugate=
false)) + sqrt(BilinearForm(v1-B2,v1-B2,conjugate=false));
```

$$v1 := \begin{bmatrix} \sqrt{2} \\ 1 \end{bmatrix}$$

$$q(v1) = 1$$

$$Abstandssumme = 4 \qquad \textbf{(2.2.9)}$$

```
> v2:=Vector([5/4,(1/8)*sqrt(78)]);
'q(v2)'=subs(map(i->y[i]=v2[i],[1,2]),q);
'Abstandssumme'=sqrt(BilinearForm(v2-B1,v2-B1,conjugate=
false)) + sqrt(BilinearForm(v2-B2,v2-B2,conjugate=false));
simplify(%);
```

$$v2 := \begin{bmatrix} \frac{5}{4} \\ \frac{1}{8}\sqrt{78} \end{bmatrix}$$

$$q(v2) = 1$$

$$Abstandssumme = \frac{1}{8}\sqrt{78 + 64\left(\frac{5}{4} - \sqrt{2}\right)^2} + \frac{1}{8}\sqrt{78 + 64\left(\frac{5}{4} + \sqrt{2}\right)^2}$$

$$Abstandssumme = 4 \tag{2.2.10}$$

Dieses erste Experimentieren stärkt unsere Vermutung, dass die Fasern Ellipsen sind. Wir wollen sie nun beweisen.
Sei dafür w ein beliebiger Punkt der Faser/Ellipse.

```
> solve(subs(y[1]=x,q)=1,y[2]);
```

$$\frac{1}{2}\sqrt{-2x^2+8},\ -\frac{1}{2}\sqrt{-2x^2+8} \tag{2.2.11}$$

Wir können also beliebige Werte für x aus dem Intervall $[-2, 2]$ wählen.
(**DENKANSTOSS**: Warum?)

```
> w:=Vector(2,[x,(1/2)*sqrt(-2*x^2+8)]);
  'q(w)'=subs(map(i->y[i]=w[i],[1,2]),q);
  ws:=Vector(2,[x,-(1/2)*sqrt(-2*x^2+8)]);
  'q(ws)'=subs(map(i->y[i]=ws[i],[1,2]),q);
```

$$w := \begin{bmatrix} x \\ \frac{1}{2}\sqrt{-2x^2+8} \end{bmatrix}$$

$$q(w) = 1$$

$$ws := \begin{bmatrix} x \\ -\frac{1}{2}\sqrt{-2x^2+8} \end{bmatrix}$$

$$q(ws) = 1 \tag{2.2.12}$$

Wir betrachten die Formel, welche die Summe der Abstände ausrechnet. Wir betrachten nur **w**, da das Vorzeichen durch Quadrieren verschwindet.

```
> sqrt(BilinearForm(w-B1,w-B1,conjugate=false)) + sqrt
  (BilinearForm(w-B2,w-B2,conjugate=false))=4;
```

$$\frac{1}{2}\sqrt{4\left(x-\sqrt{2}\right)^2 - 2x^2 + 8} + \frac{1}{2}\sqrt{4\left(x+\sqrt{2}\right)^2 - 2x^2 + 8} = 4 \tag{2.2.13}$$

Wir stellen etwas um:

```
> sqrt(BilinearForm(w-B2,w-B2,conjugate=false))=4- sqrt
  (BilinearForm(w-B1,w-B1,conjugate=false));
```

$$\frac{1}{2}\sqrt{4\left(x+\sqrt{2}\right)^2-2x^2+8}=4-\frac{1}{2}\sqrt{4\left(x-\sqrt{2}\right)^2-2x^2+8} \qquad \textbf{(2.2.14)}$$

Nun dürfen wir beide Seiten quadrieren, da die Wurzeln beide positiv sind (Abstände) und damit auch beide Seiten positiv sind.

```
> sqrt(BilinearForm(w-B2,w-B2,conjugate=false))^2=(4- sqrt
  (BilinearForm(w-B1,w-B1,conjugate=false)))^2;
  simplify(expand(%));
```

$$\left(x+\sqrt{2}\right)^2-\frac{1}{2}x^2+2=\left(4-\frac{1}{2}\sqrt{4\left(x-\sqrt{2}\right)^2-2x^2+8}\right)^2$$

$$\frac{1}{2}x^2+2x\sqrt{2}+4=20-4\sqrt{16-8x\sqrt{2}+2x^2}-2x\sqrt{2}+\frac{1}{2}x^2 \qquad \textbf{(2.2.15)}$$

Wir müssen Maple beim Vereinfachen der Gleichung unter die Arme greifen:

```
> sqrt(16-8*x*sqrt(2)+2*x^2)=4-x*sqrt(2);
```

$$\sqrt{16-8x\sqrt{2}+2x^2}=4-x\sqrt{2} \qquad \textbf{(2.2.16)}$$

Abermals dürfen wir quadrieren, da x aus dem Intervall $[-2, 2]$ war.

```
> sqrt(16-8*x*sqrt(2)+2*x^2)^2=(4-x*sqrt(2))^2;
```

$$16-8x\sqrt{2}+2x^2=\left(4-x\sqrt{2}\right)^2 \qquad \textbf{(2.2.17)}$$

```
> expand(%);
```

$$16-8x\sqrt{2}+2x^2=16-8x\sqrt{2}+2x^2 \qquad \textbf{(2.2.18)}$$

Nun erkennen wir die wahre Aussage $0=0$ und damit folgt unsere Behauptung.

ÜBUNG [08]:

1) Folgere mit Hilfe der obigen Rechnungen, dass $p^{-1}(\{1\})$ ebenfalls eine Ellipse ist, wobei p das Polynom war, welches wir in q transformiert hatten.
Hinweis: Wie muss man ein Element aus der Faser von q transformieren, um eines aus der Faser von p zu erhalten?
Hinweis: Warum ist es wichtig, dass die Transformation orthogonal war?
2) Bestimme die beiden Brennpunkte und den Wert k der Ellipse $p^{-1}(\{1\})$.
3) Plotte die beiden Ellipsen in ein gemeinsames Bild.

MATH: Eine Ellipse lässt sich immer auf die folgende Form bringen, wobei $a \geq b > 0$ ist:

```
> q_ellipse:=(y[1]/a)^2+(y[2]/b)^2;
```

$$q\_ellipse := \frac{y_1^2}{a^2}+\frac{y_2^2}{b^2} \qquad \textbf{(2.2.19)}$$

Dies wird im später im Semester bei den affinen Quadriken behandelt werden.

ÜBUNG [09]:

Zeige, dass die beiden Punkte **C1, C2** die Brennpunkte einer Ellipse in der Gestalt **q_ellipse** sind und dass die Abstandssumme k gleich $2 \cdot a$ ist.

```
> C1:=Vector([sqrt(a^2-b^2),0]);
  C2:=Vector([-sqrt(a^2-b^2),0]);
```

$$C1 := \begin{bmatrix} \sqrt{a^2 - b^2} \\ 0 \end{bmatrix}$$

$$C2 := \begin{bmatrix} -\sqrt{a^2 - b^2} \\ 0 \end{bmatrix} \qquad \textbf{(2.2.20)}$$

Hinweis: Mache dir klar, wann und warum du quadrieren darfst.

PROJEKT: Eine Hyperbel ist der geometrische Ort aller Punkte, deren Abstandsdifferenzbetrag von zwei festen Punkten einen festen Wert hat. Man kann jede Hyperbel auf die Gestalt $\left(\frac{x}{a}\right)^2 - \left(\frac{y}{a}\right)^2 = 1$ mit $0 < b \le a$ bringen.

Diskutiere die Hyperbel analog zur Ellipse.